```
/*Evaluation of FLuorometholone as Adjunctive MEdical Therapy for Trachomatous Trichiasis Surgery: The FLAME Randomized
Controlled Clinical Trial*/
/*purpose: summary baseline participant characteristics by randomized treatment groups*/
/*Note: raw datasets in SAS library 'in' corresponds to the shared data file in xlsx format, their relationship are as below:
in.eligibility2 - EL
in.surgery - SI
in.osdi_baseline - baseline record in OSI
in.Baselineinfo - DS
in.eq5d_baseline - EQ*/

/*set up SAS library, TLF path, TLF title, format catalog which will be used in later analysis*/
%include "setup.sas";
libname in "&_root_in";
libname fm "&_root_fm";
libname data "&_root_data";
%let TLF_path = &_root_output\tb1_baseline_characteristics.rtf;
%let TLF_title = Table 1:  Baseline characteristics of participants by randomized treatment group;

OPTIONS FMTSEARCH=(fm.fmrand fm.fmsurgery fm.fmelig fm.fmbaseline fm.fmosdi fm.fmeq);
option mprint;

/*set up variable format for visualization*/
proc format;
        value gf  0 = "Placebo" 1 = "FML";
        value $sexf 'm' = "Male" 'f' = "Female";
        value sexo 1='Male'
                                2='Female';
```

```
value ynf 0 = "No" 1 = "Yes";
value yno 1='No' 2='Yes';
value occuf 0 = "No job"
                    1 = "Homemaker"
                    2 = "Mainly farmer"
                    3 = "Other";
value occuo 1="No job"
                     2="Homemaker"
                     3="Mainly farmer"
                     4="Other";
value eq5df 1='None'
                  2='Slight'
                  3='Moderate'
                  4='Severe'
                  5='Extreme';
value eq5do 1='None'
                     2='Slight'
                     3='Moderate'
                     4='Severe'
                     5='Extreme';
value $eyef 'Unilateral'='Unilateral'
                  'Bilateral'='Bilateral';
value eyeo 1='Unilateral'
                  2='Bilateral';
invalue tab_order 'age'=1
                                 'sex_n'=2
                                 'eye_n'=3
```

```
                                        'scread'=4
                                        'scoccu'=5
                                        'eqpain_n'=6
                                        'osi_symp'=7
                                        'calc_osdi'=8;
        value pval (default=8)
                low - <0.00095 = '<0.001'
                0.00095 - <0.0095 = [8.3]
                0.0095 - <0.045 = [8.2]
                0.045 - <0.0495 = [8.3]
                0.0495 - <0.04995 = [8.4]
                0.04995 - <0.05 = '~<0.05'
                0.05 = '0.05'
                0.05< - <0.05005 = '~>0.05'
                0.05005 - <0.0505 = [8.4]
                0.0505 - <0.055 = [8.3]
                0.055 - 0.99 = [8.2]
                0.99< - high = [8.2];
run;

* population setting;
* data.rand includes the enrolled and randomized patient;
* data.FAS (Full analysis set) includes the participants who received TT surgery;
proc sql;
create table data.rand as
select a.subjectid as subjid, a.elelig_l as leye_eligibel, a.elelig_r as reye_eligibel,
        c.sisurgery_rul, c.sisurgery_lul,
```

```
            case when a.subjectid^='02-0071' then b.treat
                        else 1 end as treat format=gf.
            /*02-0071 has protocol deviation, actual treatment is placebo, and planned treatment is FML*/
            /*Analyses comparing treatment groups will follow the intention-to-treat (ITT) principle;
                that is, all patients will be analyzed in the group to which they are assigned regardless
                of the compliance of using the fluorometholone eye drops.*/
from in.eligibility2(where=(elelig_l=1 or elelig_r=1)) as a
left join in.id_group as b
on a.subjectid=b.subjid
left join in.surgery as c
on a.subjectid=c.subjid;
quit;
data data.rand;
        set data.rand;
        if sisurgery_rul=. then sisurgery_rul=9;
        if sisurgery_lul=. then sisurgery_lul=9;
run;
data data.fas;
        set data.rand;
        if sisurgery_rul in (0,1) or sisurgery_lul in (0,1);
run;

* derive OSDI score - OSDI ocular symptoms score, OSDI total score ;
%proc_OSDI(indata=in.osdi_baseline,outdata=OSDI_score) ;
* combine covariate used for analysis;
proc sql;
create table data.tbl1_data as
```

```
        select root.subjid, root.treat,
                case when root.sisurgery_rul not in (0,1) and root.sisurgery_lul not in (0,1) then ''
                        when root.sisurgery_rul not in (0,1) or root.sisurgery_lul not in (0,1) then 'Unilateral'
                        when root.sisurgery_rul in (0,1) and root.sisurgery_lul in (0,1) then 'Bilateral' end as eye,
                a.age, a.sex,
                case when a.scread in (1,2) then 1
                        else a.scread end as scread,
                case when a.scoccu not in (.,0,1,2) then 3
                        else a.scoccu end as scoccu,
                b.eqpain,c.osi_symp,c.calc_osdi
        from data.rand as root
        left join in.Baselineinfo as a
        on root.subjid=a.subjid
        left join in.eq5d_baseline as b
        on root.subjid=b.subjid
        left join OSDI_score as c
        on root.subjid=c.subjid;
quit;

* # of participants in each group;
proc sql noprint;
select count(*) into :t0_pat
                from data.rand
                where treat = 0;
select count(*) into :t1_pat
                from data.rand
                where treat = 1;
```

```
quit;
%put &t0_pat;
%put &t1_pat;

* descriptive statistics, hypothesis test and model building;
%stat_num(input=data.tbl1_data,output=tab_1,variable=age,cohort=treat,ref=Placebo,questionlabel=,tab_order=1,id=subjid,
gee_linear=0,pois=0,ttest=1,anova=0,wilcoxon_rank_sum=0,kruskal_wallis=0);
%stat_char(input=data.tbl1_data,output=tab_2,unit=pat,variable=sex,min=1,max=2,cohort=treat,ref=Placebo,questionlabel=
,tab_order=2,value_label=$sexf,order_label=sexo,id=subjid,gee_binomial=0,gee_multinomial=0,reverse=0,chi=1,fisher=0);
%stat_char(input=data.tbl1_data,output=tab_3,unit=pat,variable=eye,min=1,max=2,cohort=treat,ref=Placebo,questionlabel=
%str(Laterality of TT surgery
(%%)),tab_order=3,value_label=$eyef,order_label=eyeo,id=subjid,gee_binomial=0,gee_multinomial=0,reverse=0,chi=1,fisher=
0);
%stat_char(input=data.tbl1_data,output=tab_4,unit=pat,variable=scread,min=1,max=2,cohort=treat,ref=Placebo,questionla
bel=,tab_order=4,value_label=ynf,order_label=yno,id=subjid,gee_binomial=0,gee_multinomial=0,reverse=0,chi=1,fisher=0);
%stat_char(input=data.tbl1_data,output=tab_5,unit=pat,variable=scoccu,min=1,max=4,cohort=treat,ref=Placebo,questionla
bel=%str(Occupation
(%%)),tab_order=5,value_label=occuf,order_label=occuo,id=subjid,gee_binomial=0,gee_multinomial=0,reverse=0,chi=1,fishe
r=0);
%stat_char(input=data.tbl1_data,output=tab_6,unit=pat,variable=eqpain,min=1,max=5,cohort=treat,ref=Placebo,questionla
bel=%str(Pain or discomfort reported in EQ-5D
(%%)),tab_order=6,value_label=eq5df,order_label=eq5do,id=subjid,gee_binomial=0,gee_multinomial=0,reverse=0,chi=1,fishe
r=0);
%stat_num(input=data.tbl1_data,output=tab_7,variable=osi_symp,cohort=treat,ref=Placebo,questionlabel=,tab_order=7,id=
subjid,gee_linear=0,pois=0,ttest=1,anova=0,wilcoxon_rank_sum=0,kruskal_wallis=0);
%stat_num(input=data.tbl1_data,output=tab_8,variable=calc_osdi,cohort=treat,ref=Placebo,questionlabel=,tab_order=8,id=
subjid,gee_linear=0,pois=0,ttest=1,anova=0,wilcoxon_rank_sum=0,kruskal_wallis=0);
```

```
data tbl1_result;
	format answerlabel $200.;
	set tab_1(in=a) tab_2(in=b) tab_3(in=c) tab_4(in=d) tab_5(in=e) tab_6(in=f) tab_7(in=g) tab_8(in=h);
	if (a or g or h) and answerlabel^='Mean (SD)' then delete;
	if b and index(upcase(answerlabel),'FEMALE')=0 then delete;
	if d and index(upcase(answerlabel),'NO')=0 then delete;
	if a then answerlabel='Age in years: mean (SD)';
	if b then answerlabel='Gender: female (%)';
	if d then answerlabel='Reading ability: unable (%)';
	if g then answerlabel='OSDI ocular symptoms score: mean (SD)';
	if h then answerlabel='OSDI total score: mean (SD)';
run;

/*SMD calculation*/
data tbl1_data_;
	set data.tbl1_data;
	fakewt=1;
	if sex='m' then sex_n=0;
	else if sex='f' then sex_n=1;
	if eye='Unilateral' then eye_n=0;
	else if eye='Bilateral' then eye_n=1;
	eqpain_n=eqpain-1;
run;
/*for continuous variables, check skewness to see if need rank-based mean and SD to calculate stddif*/
/*Positive Skewness: if skewness > 0, data is positively skewed, otherwise negative*/
/*if skewness < -1 or >+1, the distribution is highly skewed
```

```
if skewness is between -1 and -0.5 or between 0.5 and +1, the distribution is moderately skewed
if skewness > -0.5 and < 0.5, the distribution is approximately symmetric or normal*/
proc univariate data=tbl1_data_;
        class treat;
        var age osi_symp calc_osdi;
        ods select moments;
run;
proc sgplot data=tbl1_data_;
histogram age / group=treat transparency=0.5;
run;
proc sgplot data=tbl1_data_;
histogram osi_symp / group=treat transparency=0.5;
run;
proc sgplot data=tbl1_data_;
histogram calc_osdi / group=treat transparency=0.5;
run;

%stddiff( inds = tbl1_data_,
                groupvar = treat,
                numvars = age osi_symp calc_osdi,
                charvars = sex_n eye_n scread scoccu eqpain_n,
                wtvar = fakewt,
                stdfmt = 10.5,
                outds = stddiff_result );
data stddiff_result;
        set stddiff_result;
        tab_order=input(varname,tab_order.);
```

```
run;

proc sql;
	create table data.tbl1_result as
	select a.*,
		put(round(input(b.Stddiff,best.),0.01),8.2) as Stddiff
	from tbl1_result as a
	left join stddiff_result as b
	on a.tab_order=b.tab_order
	order by a.tab_order, a.answer_order;
quit;

/*generate table*/
options nodate nonumber;
ods rtf file = "&TLF_path";
ods escapechar='~';
ods rtf text="~S={just=l font_size=10pt font_weight= bold} &TLF_title";
proc report data=data.tbl1_result nowd spanrows split='|' missing style(column)={background=white fontsize=9pt}

style(header)={background=white fontsize=9pt fontweight=medium};
	columns tab_order answer_order answerlabel cohort_0 cohort_1 /*pvalue*/ Stddiff;
	define tab_order  / ' ' order order=internal noprint;
	define answer_order  / ' ' order order=internal noprint;
	define answerlabel / 'Baseline participant characteristics' left style(column)={cellwidth=2.3in asis=ON}
style(header)={just=l};
	define cohort_0 / "Placebo|(N=%trim(&t0_pat) participants)" center style(column)={cellwidth=1.9in};
	define cohort_1 / "FLuorometholone|(N=%trim(&t1_pat) participants)" center style(column)={cellwidth=1.9in};
```

```
        *define pvalue / 'P-value' center group style(column)={cellwidth=1.2in};
        define Stddiff / 'SMD' center group style(column)={cellwidth=1.5in};
run ;
ods rtf text="~S={just=l font_size=9pt} SD=Standard deviation; TT=trachomatous trichiasis; OSDI=Ocular surface disease
index.";
*ods rtf text="~S={just=l} *Two-sample t-test for comparison of means, chi-squared test for comparison of proportions,
wilcoxon rank sum tests for skewed data.";
ods rtf text="~S={just=l font_size=9pt} SMD=Standardized Mean Difference.";
ods rtf close;

libname in clear;
libname fm clear;

proc datasets library=work nolist kill;
quit;
```